<!--#include file="./load.inc"-->
<%
viewID = indexViewID
dim qid, query, url, html, iid
qid = lcase(trim(request("q")))

if isHtml then '如果是真正伪静态 在根目录的web.config里设置
	iid = app.int(trim(request("i")))
else
	query = lcase(request.serverVariables("QUERY_STRING"))
	url = app.isTest(query, "^\/(g|l|p|a|b|o)+-(\d+)-(\d+)" & replace(fileExt, ".", "\.") & "([&|?]*?)*$")
	if isArray(url) then
		if ubound(url) > 1 then
			qid = url(0) '第1参数 数据类型
			iid = app.int(url(1)) '第2参数 信息ID
			page = app.int(url(2)) '第3参数 页码
		end if
	end if
end if

if iid > 0 then
	select case qid
		case "g" '频道首页 g-1-1.html
			call app.setGenre(iid)
			if genreID > 0 then viewID = genreViewID
		case "l" '栏目页 l-1-1.html
			call app.setClass(iid)
			if genreID > 0 then
				call app.setGenre(genreID)
				viewID = classViewID
			end if
		case "a" '新闻内容 a-1-1.html
			call app.getNews(iid)
		case "b" '留言内容 b-1-1.html
			call app.getBook(iid)
		case "p" '产品内容 p-1-1.html
			call app.getProduct(iid)
		case "o" '简介内容 o-1-1.html
			call app.getOur(iid)
	end select
	pageUrl = qid & "-" & iid 'pageUrl传参数到分页里
	if isHtml = false then pageUrl = siteUrl & "?/" & pageUrl
end if

html = app.getViewIdFile(viewID) '取得模板文件
html = app.reView(html) '替换模板标签
html = app.contentTag(html,1) '替换内容里的标签
response.write html '输出内容

call db.close() '关闭数据库
%>